# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=perl-net-frame-dump
_pkgname=Net-Frame-Dump
pkgver=1.18
pkgrel=7
pkgdesc='Base-class for a tcpdump like implementation.'
arch=('any')
url='https://metacpan.org/pod/Net::Frame::Dump'
license=('GPL' 'PerlArtistic')
depends=('perl')
options=(!emptydirs)
source=("http://search.cpan.org/CPAN/authors/id/G/GO/GOMOR/$_pkgname-$pkgver.tar.gz")
sha512sums=('ab7c826e61d303c3528d64811a4fe41c9c576b5da19eb0d9c667b00e37fd457fe3660aaff90460f3684fb5bda7acc33d8b843ec4b2e3cf5aa17aef2e15d65b28')

prepare() {
  cd "$_pkgname-$pkgver"

  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
}

build() {
  cd "$_pkgname-$pkgver"

  make
}

package() {
  cd "$_pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
}

